

NAG Library Routine Document

G05RHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G05RHF generates pseudorandom uniform variates with joint distribution of a Clayton/Cook–Johnson Archimedean copula.

2 Specification

```
SUBROUTINE G05RHF (N, M, THETA, SORDER, STATE, X, LDX, SDX, IFAIL)
  INTEGER          N, M, SORDER, STATE(*), LDX, SDX, IFAIL
  REAL (KIND=nag_wp) THETA, X(LDX,SDX)
```

3 Description

Generates n pseudorandom uniform m -variates whose joint distribution is the Clayton/Cook–Johnson Archimedean copula C_θ , given by

$$C_\theta = (u_1^{-\theta} + u_2^{-\theta} + \dots + u_m^{-\theta} - m + 1)^{-1/\theta}, \quad \begin{cases} \theta \in (0, \infty), \\ u_j \in (0, 1], \quad j = 1, \dots, m; \end{cases}$$

with the special case:

$$C_\infty = \min(u_1, u_2, \dots, u_m), \text{ the Fréchet–Hoeffding upper bound.}$$

The generation method uses mixture of powers.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05RHF.

4 References

Marshall A W and Olkin I (1988) Families of multivariate distributions *Journal of the American Statistical Association* **83** 403

Nelsen R B (2006) *An Introduction to Copulas* (2nd Edition) Springer Series in Statistics

5 Arguments

- | | | |
|----|---|--------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of pseudorandom uniform variates to generate. | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | M – INTEGER | <i>Input</i> |
| | <i>On entry:</i> m , the number of dimensions. | |
| | <i>Constraint:</i> $M \geq 2$. | |
| 3: | THETA – REAL (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> θ , the copula parameter. | |
| | <i>Constraint:</i> $\text{THETA} \geq 1.0 \times 10^{-6}$. | |

- 4: SORDER – INTEGER *Input*
On entry: determines the storage order of variates; the (i, j) th variate is stored in $X(i, j)$ if $SORDER = 1$, and $X(j, i)$ if $SORDER = 2$, for $i = 1, 2, \dots, n$ and $j = 1, 2, \dots, m$.
Constraint: $SORDER = 1$ or 2 .
- 5: STATE(*) – INTEGER array *Communication Array*
Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 6: X(LDX, SDX) – REAL (KIND=nag_wp) array *Output*
On exit: the pseudorandom uniform variates with joint distribution described by C_θ , with $X(i, j)$ holding the i th value for the j th dimension if $SORDER = 1$ and the j th value for the i th dimension of $SORDER = 2$.
- 7: LDX – INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G05RHF is called.
Constraints:
 if $SORDER = 1$, $LDX \geq N$;
 if $SORDER = 2$, $LDX \geq M$.
- 8: SDX – INTEGER *Input*
On entry: the second dimension of the array X as declared in the (sub)program from which G05RHF is called.
Constraints:
 if $SORDER = 1$, $SDX \geq M$;
 if $SORDER = 2$, $SDX \geq N$.
- 9: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1 . If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: $IFAIL = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, corrupt STATE argument.

IFAIL = 2

On entry, invalid THETA: THETA = $\langle value \rangle$.
Constraint: THETA $\geq 1.0 \times 10^{-6}$.

IFAIL = 3

On entry, N = $\langle value \rangle$.
Constraint: N ≥ 0 .

IFAIL = 4

On entry, M = $\langle value \rangle$.
Constraint: M ≥ 2 .

IFAIL = 5

On entry, invalid SORDER.
Constraint: SORDER = 1 or 2.

IFAIL = 7

On entry, LDX must be at least $\langle value \rangle$: LDX = $\langle value \rangle$.

IFAIL = 8

On entry, SDX must be at least $\langle value \rangle$: SDX = $\langle value \rangle$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.
See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05RHF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

In practice, the need for numerical stability restricts the range of θ such that:

the routine requires $\theta \geq 1.0 \times 10^{-6}$;

if $\theta > 200.0$, the routine returns pseudorandom uniform variates with C_∞ joint distribution.

10 Example

This example generates thirteen four-dimensional variates for copula $C_{1,3}$.

10.1 Program Text

```

Program g05rhfe

!      G05RHF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: g05kff, g05rhf, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter                :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)                :: theta
Integer                            :: genid, ifail, ldx, lstate, m, n,      &
                                   sdx, sorder, subid
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable   :: x(:, :)
Integer                            :: seed(lseed)
Integer, Allocatable               :: state(:)
!      .. Executable Statements ..
Write (nout,*) 'G05RHF Example Program Results'
Write (nout,*)
Flush (nout)

!      Skip heading in data file
Read (nin,*)

!      Read in the base generator information and seed
Read (nin,*) genid, subid, seed(1)

!      Initial call to initializer to get size of STATE array
lstate = 0
Allocate (state(lstate))
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Reallocate STATE
Deallocate (state)
Allocate (state(lstate))

!      Initialize the generator to a repeatable sequence
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Read in sample size, number of dimensions and order
Read (nin,*) n, m, sorder

If (sorder==1) Then
!      X(N,M)
    ldx = n
    sdx = m
Else
!      X(M,N)
    ldx = m
    sdx = n
End If
Allocate (x(ldx,sdx))

```

```

!      Read in parameter
      Read (nin,*) theta

!      Generate variates
      ifail = 0
      Call g05rhf(n,m,theta,sorder,state,x,ldx,sdx,ifail)

!      Display the variates
      If (sorder==1) Then
!         X(N,M)
         ifail = 0
         Call x04caf('General',' ',n,m,x,ldx,                &
           'Uniform variates with copula joint distribution',ifail)
      Else
!         X(M,N)
         ifail = 0
         Call x04caf('General',' ',m,n,x,ldx,                &
           'Uniform variates with copula joint distribution',ifail)
      End If

      End Program g05rhfe

```

10.2 Program Data

```

G05RHF Example Program Data
1 1 1762543          :: GENID,SUBID,SEED(1)
13 4 1              :: N,M,SORDER
1.3                 :: THETA

```

10.3 Program Results

G05RHF Example Program Results

Uniform variates with copula joint distribution

	1	2	3	4
1	0.8576	0.5048	0.9761	0.5895
2	0.3186	0.6372	0.9959	0.5898
3	0.9050	0.6950	0.9353	0.9329
4	0.5278	0.1804	0.4177	0.2330
5	0.1510	0.9777	0.2621	0.3867
6	0.3906	0.7938	0.3073	0.3150
7	0.1279	0.1709	0.1751	0.0568
8	0.7613	0.4314	0.3498	0.2913
9	0.3871	0.4430	0.3610	0.3774
10	0.1242	0.0647	0.0472	0.0780
11	0.6866	0.9500	0.9289	0.9763
12	0.5259	0.8218	0.7134	0.4914
13	0.0955	0.0459	0.1265	0.1947
